Lie Algebra methods for Accelerator Physics in 2D

1. Poisson Brackets, Lie Operator and Exponential Lie Operators

In[ ]:=
```
PoissonBracket[f_, g_, q_Symbol, p_Symbol] :=
 Simplify[D[f, q] * D[g, p] - D[f, p] * D[g, q]]
```

In[ ]:=
```
PoissonBracket[f[x, p], g[x, p], x, p]
```

Out[ ]=

$$g^{(0,1)}[x, p]\, f^{(1,0)}[x, p] - f^{(0,1)}[x, p]\, g^{(1,0)}[x, p]$$

In[ ]:=
```
LieOperator[f_] := Function[g, PoissonBracket[f, g, x, p]]
```

In[ ]:=
```
LieOperator[f[x, p]][g[x, p]]
```

Out[ ]=

$$g^{(0,1)}[x, p]\, f^{(1,0)}[x, p] - f^{(0,1)}[x, p]\, g^{(1,0)}[x, p]$$

In[ ]:=
```
ExpOperator[oprtr_, n_ : 10] :=
 Function[f, Fold[f + oprtr[#1] / #2 &, f, Reverse@Range[n]]]
```

In[ ]:=
```
DiffOperator := Function[f, D[f, x]]
```

In[ ]:=
```
Simplify[ExpOperator[DiffOperator, 4][f[x]]]
```

Out[ ]=

$$f[x] + f'[x] + \frac{f''[x]}{2} + \frac{1}{6} f^{(3)}[x] + \frac{1}{24} f^{(4)}[x]$$

In[ ]:=
```
ExpOperator[LieOperator[f[x, p]], 1][g[x, p]]
```

Out[ ]=

$$g[x, p] + g^{(0,1)}[x, p]\, f^{(1,0)}[x, p] - f^{(0,1)}[x, p]\, g^{(1,0)}[x, p]$$

2. Lie Methods for common accelerator elements

In[ ]:=
```
Drift[L_, n_ : 10] := Function[g, ExpOperator[LieOperator[-0.5 * L * p^2], n][g]]
```

In[∘]:=
```
Map[Drift[L], {x, p}]
```
Out[∘]=

$\{0. + 1.\ L\ p + x,\ 0. + p\}$

In[∘]:=
```
ThinQuad[kL_, n_ : 10] := Function[g, ExpOperator[LieOperator[-0.5 * kL * x^2], n][g]]
```

In[∘]:=
```
Map[ThinQuad[kL], {x, p}]
```
Out[∘]=

$\{0. + x,\ 0. + p - 1.\ kL\ x\}$

In[∘]:=
```
ThinKick[k_, m_, n_ : 10 ] := Function[g, ExpOperator[LieOperator[k * x^m], n][g]]
```

In[∘]:=
```
Map[ThinKick[λ, n], {x, p}]
```
Out[∘]=

$\{x,\ p + n\ x^{-1+n}\ \lambda\}$

In[∘]:=
```
ThickQuad[k_, L_, n_ : 10] := Function[g,
  ExpOperator[LieOperator[-Sign[k] * 0.5 * L * (k * x^2 + Sign[k] * p^2)], n][g]]
```

In[∘]:=
```
Collect[Assuming[k < 0, ThickQuad[k, L, 3][x]], {x, p}]
```
Out[∘]=

$L\ (1. + 0.166667\ k\ L^2)\ p + (1 + 0.5\ k\ L^2)\ x$

In[∘]:=
```
Collect[Assuming[k < 0, ThickQuad[k, L, 3][p]], {x, p}]
```
Out[∘]=

$(1 + 0.5\ k\ L^2)\ p + (1.\ k\ L + 0.166667\ k^2\ L^3)\ x$

In[∘]:=
```
ThickFQuad[k_, L_, n_ : 10] :=
 Function[g, ExpOperator[LieOperator[-0.5 * L * (k * x^2 + p^2)], n][g]]
```

In[∘]:=
```
Collect[ThickFQuad[k, L, 10][p], {x, p}]
```
Out[∘]=

$$(1 - 0.5\ k\ L^2 - 2.75573\times 10^{-7}\ k^2\ L^3\ (-151\,200.\ L + 5040.\ k\ L^3 - 90.\ k^2\ L^5 + 1.\ k^3\ L^7))\ p +$$
$$(-1.\ k\ L - 2.75573\times 10^{-6}\ k^2\ L^3\ (-60\,480. + 3024.\ k\ L^2 - 72.\ k^2\ L^4 + 1.\ k^3\ L^6))\ x$$

In[∘]:=
```
ThickDQuad[k_, L_, n_ : 10] :=
 Function[g, ExpOperator[LieOperator[0.5 * L * (k * x^2 - p^2)], n][g]]
```

In[∘]:=
```
Collect[ThickDQuad[k, L, 6][p], {x, p}]
```
Out[∘]=

$$(1 + 0.5\ k\ L^2 + 0.0416667\ k^2\ L^4 + 0.00138889\ k^3\ L^6)\ p +$$
$$(1.\ k\ L + 0.166667\ k^2\ L^3 + 0.00833333\ k^3\ L^5)\ x$$

In[∘]:=
```
Collect[ExpOperator[LieOperator[(0.5 * p^2 + k * x^2) * s]][x], {x, p}]
```
Out[∘]=

$$\frac{1}{3}\ p\ s\ (-3. + 1.\ k\ s^2 - 0.1\ k^2\ s^4 + 0.0047619\ k^3\ s^6 - 0.000132275\ k^4\ s^8) +$$
$$\left(1 + \frac{1}{3}\ k\ s^2\ (-3. + 0.5\ k\ s^2 - 0.0333333\ k^2\ s^4 + 0.00119048\ k^3\ s^6 - 0.000026455\ k^4\ s^8)\right)\ x$$

In[∘]:=
```
Collect[ExpOperator[LieOperator[-(0.5*p^2+k*x^3/6)*s], 3][x], {x, p}]
```
Out[∘]=

$$1.\, p\, s + \left(1 - 0.166667\, k\, p\, s^3\right) x - 0.25\, k\, s^2\, x^2$$

3. Concatenating elements by composition of maps

In[∘]:=
```
Map[Composition[Drift[L], Drift[L]], {x, p}]
```
Out[∘]=

$$\{0. + 2.\, L\, p + x,\ 0. + p\}$$

FODO Lattice

In[∘]:=
```
Collect[Simplify[Composition[ThinQuad[-1/(2 f)],
    Drift[L], ThinQuad[1/f], Drift[L], ThinQuad[-1/(2 f)]][x]], {x, p}]
```
Out[∘]=

$$0. + \left(2.\, L - \frac{1.\, L^2}{f}\right) p + \left(1 - \frac{0.5\, L^2}{f^2}\right) x$$

In[∘]:=
```
Collect[Simplify[Composition[ThinQuad[-1/(2 f)],
    Drift[L], ThinQuad[1/f], Drift[L], ThinQuad[-1/(2 f)]][p]], {x, p}]
```
Out[∘]=

$$0. + \left(1 - \frac{0.5\, L^2}{f^2}\right) p + \left(-\frac{0.5\, L}{f^2} - \frac{0.25\, L^2}{f^3}\right) x$$